```
* Program.: CASH.PRG
* Author..: Stephen Kurasch
* Date....: June 10, 1985
* Notice..: Copyright 1985, Ashton-Tate
* Notes...: Enter Cash Withdrawals and Automatic Teller transactions
            Called from CBMENU.PRG
* check to make sure balance is greater than 0.00
IF balance <= 0.00
   CLEAR
   @ 5,11 SAY "A cash withdrawal is not possible - your balance is " +;
              LTRIM(STR(balance, 10, 2))
   WAIT SPACE(19) + "Press any key to return to the Main menu "
   RETURN
ENDIF
* clear screen and draw menu
CLEAR
@ 2,0 TO 11,79 DOUBLE
@ 4,0 SAY CHR(204)+dhoriz+CHR(185)
@ 3,20 SAY "CASH WITHDRAWAL/AUTOMATIC TELLER SUBMENU"
DO WHILE .T.
   * display options
   @ 6,25 SAY " 0 - Exit"
   @ 7,25 SAY " 1 - Cash Withdrawal"
   @ 8,25 SAY " 2 - Automatic Teller"
   @ 11,33 SAY " select : : "
   * enter choice
   choice2 = "Z"
   DO WHILE .NOT. choice2 $ " 012"
      choice2 = " "
      @ 11,42 GET choice2
      READ
   ENDDO
   * put asterisk on screen next to choice
   @ 6+VAL(choice2),24 SAY "*"
   * erase from screen other options not selected
   counter = 0
   DO WHILE counter < 3
      IF counter <> VAL(choice2)
         @ 6+counter, 25 SAY SPACE(21)
      ENDIF
      counter = counter + 1
   ENDDO
   * test for exit condition - close out files and release variables
   IF choice2 $ " 0"
      @ 14,28 SAY "Returning to Main Menu"
```

```
EXIT
ENDIF
* initialize memory variables
option = IIF(choice2 = "1", "Cash Withdrawal ", "Automatic Teller ")
mdate = today
* input date
@ 14,19 SAY "Enter date of "+option GET mdate
* loop to enter positive amount
DO WHILE .T.
   mamt = 0.00
   @ 17,10 SAY "Enter amount received from "+option GET mamt
   READ
   @ 19,0 CLEAR
   DO CASE
      CASE mamt < 0.00
         @ 19,17 SAY "A positive amount is needed - please reenter"
      CASE mamt = 0.00
         @ 19,21 SAY "An amount is needed - please reenter"
      CASE mamt > balance
         @ 19,33 SAY "BANK ROBBER!!!"
         @ 20,13 SAY "You cannot withdraw more than what's "+;
                     "in your account"
      OTHERWISE
         * valid transaction
         EXIT
   ENDCASE
ENDDO
* confirm transaction
answer = " "
DO WHILE .NOT. answer$"YyNn"
   answer=" "
   @ 21,20 SAY "Is the above transaction correct? (Y/N)" GET answer
   READ
ENDDO
IF answer$"Yy"
   @ 22,16 SAY "One moment while appropriate files are updated"
   * update individual deposits file
   USE Deposits
   APPEND BLANK
   REPLACE Date WITH mdate
   REPLACE Amt WITH -mamt
   REPLACE Paidfrom WITH option
   * update memory variables to be saved in CHKBOOK.MEM
```

```
lastwth = lastwth + 1
      balance = balance - mamt
      * update bank deposits file
      USE Bank
      APPEND BLANK
      REPLACE Date WITH mdate
      REPLACE Amt WITH -mamt
      REPLACE Num WITH 0
      REPLACE Numwith WITH lastwth
      REPLACE Clear WITH .F.
      * close out file
      USE
   ENDIF
   * erase display from lower part of screen
   @ 14,0 CLEAR
   * erase asterisk from current menu display
   @ VAL(choice2)+6,24 SAY " "
ENDDO
RETURN
```

Cash.prg

\* Eof: